
Programs
1. The VARIOGRAM macro creates the data set varioplot that contains the data values to construct 
a sample variogram. The data must be sorted by the subject’s identification number.
%macro variogram (data=,resvar=,clsvar=,expvars=,id=,time=,maxtime=,);
 
ods select none;

proc mixed data=&data;
   class &clsvar;
   model &resvar=&expvars / outpm=residuals;
run;

ods select all;
An output data set is created with the residuals from the mean model.
data residuals1;
   set residuals;
   by &id;
   if first.&id then timegrp=1;
   else timegrp+1;
run;

proc transpose data=residuals1 out=subject prefix=time;
   var resid &time;
   by &id;
   id timegrp;
run;
BY-group processing is used to create timegrp, which indicates what time point (first, second, third, 
and so on) the observation is. The TRANSPOSE procedure is used to create two observations per subject with the variables time1, time2, and so on, which represent the residual values and time values.
data variogram_table(keep=variogram)
     time_interval_table(keep=time_interval);
   set subject;
   array time(*) time1-time&maxtime; 
   array diff(%eval(&maxtime-1),%eval(&maxtime-1));
   array timei(%eval(&maxtime-1),%eval(&maxtime-1));
   if _name_='Resid' then
      do i=1 to %eval(&maxtime-1);
         do k=i+1 to &maxtime;
            if time(i) ne . and time(k) ne . then
              do;
                diff(i,k-1)=((time(i)-time(k))**2)/2;
              end;
          end;
      end;
   else
     do i=1 to %eval(&maxtime-1);
        do k=i+1 to &maxtime;
           if time(i) ne . and time(k) ne . then
             do;
               timei(i,k-1)=abs(time(i)-time(k));
             end;
        end;
     end;
  do i=1 to %eval(&maxtime-1);
     do k=i to %eval(&maxtime-1);
          if diff(i,k) ne . then
              do; 
                variogram=diff(i,k);
                  output variogram_table;
              end;
        else
          if timei(i,k) ne . then
                do;
              time_interval=timei(i,k);
                    output time_interval_table;
                end;
     end;
  end;
run;
The first set of DO loops calculates the observed half-squared differences between pairs of residuals within each subject.
The second set of DO loops calculates the corresponding time differences within each subject.
The third set of DO loops outputs two data sets. The data set variogram_table has the variogram values (observed half-squared differences between pairs of residuals). The data set time_interval has the time interval values (corresponding time differences).
data varioplot;
   merge variogram_table time_interval_table;
run;

%mend variogram;
The DATA step does a one-to-one merge of the two data sets. This is appropriate because the data sets are in identical order.
2. The VARIANCE macro computes the variogram-based estimate of the process variance. The data must be sorted by the subject’s identification number.
%macro variance(data=,id=,resvar=,clsvar=,expvars=,
                subjects=,maxtime=,);

ods select none;

proc mixed data=&data;
   class &clsvar;
   model &resvar=&expvars / outpm=residuals;
run;

ods select all;
An output data set is created with the residuals from the mean model.
data residvar;
   set residuals;
   by &id;
   if first.&id then timegrp=1;
   else timegrp+1;
run;

proc transpose data=residvar out=varsubject
               prefix=time;
   var resid;
   by &id;
   id timegrp;
run;
BY-group processing is used to create timegrp, which indicates what time point (first, second, third, and so on) the observation is. The TRANSPOSE procedure is used to create one observation per subject with the variables time1, time2, and so on, representing the residual values.
data variance1(keep=diff1-diff%eval(&maxtime*&subjects)); 
   retain timepts1-timepts%eval(&maxtime*(&subjects+1))
       diff1-diff%eval(&maxtime*&subjects);
   set varsubject end=lastone;
   array time{&maxtime};
   array timepts{%eval(&subjects+1),&maxtime};
   array diff(&subjects,&maxtime);
  do i=1 to &maxtime; 
     timepts(_n_,i)=time(i);
  end;
  if lastone=1 then
    do;
      do i=1 to &subjects;
         do j=1 to &maxtime;
            do k=1 to %eval(&subjects+1)-i;
               do l=1 to &maxtime;
                  diff(k,l)=((timepts(i,j) - 
                              timepts(k+i,l))**2)*1/2;
               end;
            end;
            output;
            do k=1 to &subjects;
              do l=1 to &maxtime;
                 diff(k,l)=.;
              end;
            end;
        end;
      end;
    end;
run;
The DATA step calculates the variance by comparing each subject’s time points to all other time points for all other subjects. Three arrays are created. The time array contains the time points from the varsubject data set. The timepts array contains the time points for all the subjects from the varsubject data set. The diff array contains differences between one time point for one subject with all time points for all the other subjects.
The first DO loop reads the time points into the timepts array.
After reading in the data set varsubject, populate the array containing differences between one time point for one subject with all the time points for all the other subjects. In the DO group, the first DO loop is 
the loop for one subject. The second DO loop is the loop for each time point for one subject. The third DO loop cycles through the other subjects. The fourth DO loop cycles through all the time points for each subject.
The last two DO loops clear out the difference values to avoid carrying over any residual values.
data average_variance(keep=average total nonmissing);
  array diff{%eval(&maxtime*&subjects)};
  set variance1 end=lastone;
  nonmissing+n(of diff1-diff%eval(&maxtime*&subjects));
  total+sum(of diff1-diff%eval(&maxtime*&subjects));
  if lastone=1 then
    do;
      average=total/nonmissing;
      output;
    end;
run;
The DATA step calculates the average of all differences. It accumulates the number of nonmissing differences and the total of nonmissing differences. After reading all the difference values in the data set, it calculates the average difference and writes the average to a data set.
proc print data=average_variance;
run;

%mend variance;
3. This program computes a likelihood ratio test that compares a random coefficients model with 
the cubic effect of time with a random coefficients model with the quadratic effect of time for 
the aids data set.
ods select none; 
 
proc mixed data=sasuser.aids; 
   model cd4_scale=time age cigarettes drug partners 
         depression time*age time*depression 
         time*drug time*partners time*cigarettes 
         time*time time*time*time  / ddfm=kr;  
   random intercept time time*time time*time*time / 
          type=un subject=id; 
   ods output lrt=cubicmodel; 
run; 
  
proc mixed data=sasuser.aids; 
   model cd4_scale=time age cigarettes drug partners 
         depression time*age time*depression 
         time*drug time*partners time*cigarettes 
         time*time time*time*time  / ddfm=kr;  
   random intercept time time*time / type=un subject=id; 
   ods output lrt=quadmodel; 
run;
The output object of interest is called LRT, which has the results of the null model likelihood ratio test. The SAS data set cubicmodel has the results of the cubic model while quadmodel has the results of the quadratic model.
data likelihood; 
   merge cubicmodel (rename=(df=df3 chisq=chisq3)) 
         quadmodel (rename=(df=df2 chisq=chisq2)); 
   testcubic=chisq3-chisq2; 
   dfcubic=df3-df2; 
   pvaluecubic=1-probchi(testcubic,dfcubic); 
run; 
The variable CHISQ has the likelihood ratio test statistic while DF has the degrees of freedom. Subtracting the two test statistics and the two degrees of freedom will give the information needed to compute the likelihood ratio test comparing the two models. The PROBCHI function is used to compute the p-value for the test.
	If one of the variance components is 0, then the NOBOUND option should be used when computing the likelihood ratio test.
ods select all; 
 
proc print data=likelihood split='*' noobs; 
   var testcubic dfcubic pvaluecubic; 
   label testcubic='likelihood ratio*test 
         statistic*comparing*cubic model to quadratic 
         model' 
         dfcubic='degrees of*freedom' 
         pvaluecubic='p-value'; 
   title 'Likelihood Ratio Test for the Cubic Model';
run; 
 
                          Likelihood Ratio Test for the Cubic Model                          

                          likelihood ratio
                           test statistic
                              comparing              degrees of
                   cubic model to quadratic model      freedom        p-value

                               23.9263                    5        .000224306
The test statistic is clearly significant. Therefore, the cubic effect for time should remain in the RANDOM statement.
4. This program computes a likelihood ratio test for the fixed effects comparing the full model with a model without the three non-significant interactions for the aids data set. The program only computes the degrees of freedom correctly when there are no classification variables in the MODEL statement.
ods select none;

proc mixed data=sasuser.aids method=ml; 
   model cd4_scale=time age cigarettes drug partners 
         depression time*age time*depression 
         time*drug time*partners time*cigarettes 
         time*time time*time*time  / ddfm=kr;  
   random intercept time time*time time*time*time / 
          type=un subject=id; 
   ods output dimensions=fulldim 
              fitstatistics=fullfit; 
run; 
proc mixed data=sasuser.aids method=ml; 
   model cd4_scale=time age cigarettes drug partners 
         depression time*age 
         time*cigarettes time*time time*time*time / 
           ddfm=kr; 
   random intercept time time*time time*time*time / 
          type=un subject=id; 
   ods output dimensions=subsetdim  
              fitstatistics=subsetfit; 
run; 
The output objects of interest are DIMENSIONS, which has the number of fixed effect parameters 
in the model, and FITSTATISTICS, which has the –2 log likelihood value for the model. 
The SAS data set
· fulldim has the number of fixed effect parameters in the full model 
· fullfit has the –2 log likelihood value for the full model 
· subsetdim has the number of fixed effect parameters in the subset model 
· subsetfit has the –2 log likelihood value for the subset model 
	If a fixed effect is in the CLASS statement, then the degrees of freedom must be calculated differently.
data _null_; 
   set fulldim; 
   if descr='Columns in X'; 
   call symput('dffull',value); 
run; 
 
data _null_; 
   set subsetdim; 
   if descr='Columns in X'; 
   call symput('dfsubset',value); 
run; 
data _null_; 
   set fullfit; 
   if descr='-2 Log Likelihood'; 
   call symput('fulllr',value); 
run; 
 
data _null_; 
   set subsetfit; 
   if descr='-2 Log Likelihood'; 
   call symput('subsetlr',value); 
run; 
The four DATA _null_ steps create four macro variables with the two values of fixed-effect parameters and the two values of –2 log likelihoods.
data results; 
   testfull=&subsetlr-&fulllr; 
   dffull=&dffull-&dfsubset; 
   pvaluefull=1-probchi(testfull,dffull); 
run; 
The testfull variable has the likelihood ratio test statistic value and the dffull variable has the test statistic degrees of freedom.
ods select all; 
  
proc print data=results split='*'; 
   var testfull dffull pvaluefull; 
   label testfull='likelihood ratio*test statistic* 
                   comparing*full model to subset model' 
         dffull='degrees of*freedom' 
         pvaluefull='p-value'; 
   title 'Likelihood Ratio Test for the 3 Fixed Effect '
         'Interactions'; 
run; 
 
                  Likelihood Ratio Test for the 3 Fixed Effect Interactions                  

                               likelihood ratio
                                test statistic
                                            comparing    degrees of
                  Obs     full model to subset model       freedom     p-value

                   1                2.37334                   3        0.49862
The likelihood ratio test is clearly not significant. Therefore, the three interaction terms should be eliminated from the model.

Preisser and Qaqish (1996) proposed case-deletion regression diagnostics for the GEE methodology. The diagnostics are an approximation to the difference in the estimated regression coefficients that one would obtain upon deleting either one observation or one cluster. They proposed the computationally feasible one-step approximation diagnostics, which are similar to the ones proposed by Pregibon (1981) for generalized linear models. The authors also showed that the one-step diagnostic statistics were very good approximations to their exact diagnostic counterparts and their computations were very fast. They recommend that these diagnostic statistics be routinely used in data analysis for GEE models. However, there are no published cutoffs for these diagnostic statistics.

The leverage is simply the diagonal elements in the hat matrix, which corresponds to the amount 
of influence the observation has on the fitted values. PROC GENMOD produces leverage values for both observations and clusters.

Cook’s D statistic (Cook 1977, 1979) measures the influence of observations on the estimated values 
of the linear predictor. These diagnostic statistics measure the influence of a deleted observation or cluster on the overall fit of the model. PROC GENMOD also produces these statistics for both observations and clusters.

Another diagnostic statistic proposed by Preisser and Qaqish (1996) examines the lack of fit of a cluster. The difference between this statistic and Cook’s D is that the cluster is deleted from both the numerator and denominator. Cook’s D is useful in that the comparison of distances between clusters is meaningful because they refer to the same metric. However, because the deleted cluster influences the estimate 
of the variance, its inclusion might decrease the magnitude of the diagnostic and might hide influence. Cook (1986) points out that the studentized diagnostic (Preisser lack-of-fit statistic) has a different interpretation than the standardized version (Cook’s D). The studentized diagnostic has the interpretation of the influence of the cluster on the parameter estimates and the variance estimate of the parameter estimates simultaneously. Therefore, Preisser and Qaqish (1996) recommend that the question “influence on what?” should be the determining factor in your choice of statistics.

PROC GENMOD computes four dfbeta statistics. The statistic dfbetac is the effect of deleting a cluster, dfbetacs is the standardized version of the cluster deletion statistic, dfbetao is the effect of deleting an observation, and dfbetaos is the standardized version of the observation deletion statistic.

Example:	Fit a GEE model on the keratotomy data and specify the unstructured correlation structure 
and reference cell coding for gender with female as the reference cell. Produce the ODS Statistical Graphics on the GEE diagnostic plots by requesting the cluster leverage plot, the cluster Cook’s D plot, the cluster lack of fit plot, and the standardized cluster dfbeta plot and label the observations by the case number. Also create a SAS data set with the diagnostic statistics for both the observations and the clusters.
/* long0bd01.sas */
proc genmod data=sasuser.keratotomy desc plots(clusterlabel)=
         (cleverage dcls mcls dfbetacs);
   class patientid gender (param=ref ref='Female');
   model unstable = age diameter gender visit / dist=bin;
   repeated subject = patientid / type=unstr;
   output out=diagnostics CLEVERAGE=cleverage CLUSTER=cluster 
          CLUSTERCOOKSD=clustercooksd DFBETAC=_all_ DFBETACS=_all
          CLUSTERDFIT=clusterdfit COOKSD=cooksd LEVERAGE=leverage; 
   title 'GEE Model of Radial Keratotomy Surgery';
run;
Selected PROC GENMOD statement options:
PLOTS=	specifies plots to be created using ODS Graphics. Many of the observational statistics in the output data set can be plotted using this option. You are not required to create an output data set in order to produce a plot.
Selected global plot option:
CLUSTERLABEL	displays formatted levels of the SUBJECT= effect instead of plot symbols. This option applies only to diagnostic statistics for models fit by GEEs that are plotted against cluster number, and provides a way to identify cluster level names with corresponding ordered cluster numbers.
Selected plot options:
CLEVERAGE	plots the cluster leverage as a function of ordered cluster. 
DCLS	plots the cluster Cook’s distance statistic as a function of ordered cluster. 
MCLS	plots the studentized cluster Cook’s distance statistic as a function of ordered cluster. 
DFBETACS	plots the standardized cluster deletion statistic as a function of ordered cluster for each regression parameter in the model. 
Selected OUTPUT statement keywords:
CLEVERAGE	represents the leverage of a cluster.
CLUSTER	represents the numerical cluster index, in order of sorted clusters. 
[bookmark: DemoB_3_pg15]CLUSTERCOOKSD	represents the Cook distance-type statistic to measure the influence of deleting an entire cluster on the overall model fit. 
DFBETAC=	represents the effect of deleting an entire cluster on parameter estimates. If you specify the keyword _all_ after the equal sign, variables named DFBETAC_ParameterName will be included in the output data set to contain the values of the diagnostic statistic to measure the influence of deleting the cluster on the individual parameter estimates. ParameterName is the name of the regression model parameter formed from the input variable names concatenated with the appropriate levels, if classification variables are involved. 
DFBETACS=	represents the effect of deleting an entire cluster on normalized parameter estimates. If you specify the keyword _all_ after the equal sign, variables named DFBETACS_ParameterName will be included in the output data set to contain the values of the diagnostic statistic to measure the influence of deleting the cluster on the individual parameter estimates, normalized by their standard errors. 
CLUSTERDFIT	represents the studentized Cook distance-type statistic to measure the influence of deleting an entire cluster on the overall model fit. 
COOKSD	represents the Cook distance type statistic to measure the influence of deleting a single observation on the overall model fit. 
LEVERAGE	represents the leverage of a single observation.

[bookmark: DemoB_3_pg16]Several clusters appear to have high GEE diagnostic statistics. You should examine influential clusters and determine whether they are erroneous. If these clusters are legitimate, then they might represent important new findings. They also might indicate that your current model is inadequate.

The observation labels are difficult to see because so many observations fall in the same space in the diagnostic plot.
Example:	Print the clusters that exceed the 99th percentile for the distribution of Cook’s D for the cluster and the cluster lack of fit statistic.
proc rank data=diagnostics groups=100 out=percentiles;
   var clustercooksd clusterdfit;
   ranks percentilecooksd percentiledfit;
run;

proc print data=percentiles noobs;
   where percentilecooksd = 99 or percentiledfit = 99;
   var patientid unstable gender age diameter visit clustercooksd 
       cooksd cleverage leverage clusterdfit;
   title 'Clusters with Possible Outlying Covariate Patterns';
run;

 99   1   Female   23.2252   3.0    1   0.017523   .003971849   0.021370   .006157116   0.017214
 99   1   Female   23.2252   3.0    4   0.017523   .000278872   0.021370   .006504543   0.017214
 99   1   Female   23.2252   3.0   10   0.017523   .000366719   0.021370   .008708255   0.017214
131   0   Male     51.9042   3.5    1   0.016831   .000559362   0.026457   .008853698   0.016457
131   0   Male     51.9042   3.5    4   0.016831   .000002554   0.026457   .008896500   0.016457
131   0   Male     51.9042   3.5   10   0.016831   .006604576   0.026457   .008707161   0.016457
186   1   Female   23.1841   3.0    1   0.017579   .003982858   0.021422   .006170329   0.017269
186   1   Female   23.1841   3.0    4   0.017579   .000279700   0.021422   .006521120   0.017269
186   1   Female   23.1841   3.0   10   0.017579   .000367831   0.021422   .008730165   0.017269
The output shows which clusters had a high cluster Cook’s D or a high cluster lack of fit statistic.

Cook’s D Statistic

•

The statistic is a measure of the simultaneous change in the parameter 

estimates when an observation or cluster is deleted from the analysis.

•

The measure is scaled by the variance estimate of the parameter estimate 

based on all the observations.

•

Statistics are produced for both observations and clusters.

Cluster Fit Statistic

The cluster lack-of-fit statistic does the following: 

•

examines the lack of fit of a cluster

•

differs from Cook’s D in that the measure is scaled by the variance of the 

parameter estimate based on all but a subset of observations that defines 

the cluster.

Dfbetas

•

Dfbetas measure the change in parameter estimates when an observation 

or cluster is deleted from the analysis.

•

They are produced for both observations and clusters.

•

Both the standardized and unstandardized versions can be specified.


image7.png




image8.png
Cluster Leverage

Cluster Cook's D

GEE Diagnostic Statistics for unstable

004

0034

002

0014

000

=

1o

0015

o010

0005

0000

0 .
o 0015

m

2 = o104
2407 1
128t 13 103

1% ST
35 g, 110

Cluster DFFIT

0005

0000

Ordered Cluster

Ordered Cluster





image9.png
Standardized Cluster DFBETA Plots for unstable

g oo
:
01 %
i
248 20620 4
i 28 02
5 £ 5 o]
: £ :
L] i 0104

Ordered Cluster Ordered Cluster Ordered Cluster





image10.png
End of Demonstration




image1.emf
Copyright © SAS Institute Inc. All rights reserved.

4

Objectives

•

Define the GEE case deletion diagnostic statistics created in PROC 

GENMOD.

•

Plot the GEE case deletion diagnostic statistics.

Case Deletion Diagnostic Statistics

•

PROC GENMOD now computes GEE diagnostics, which account for the 

leverage and residuals in a set of observations to determine their influence 

on regression parameter estimates and fitted values.

•

PROC GENMOD also computes observation-deletion diagnostics and 

cluster-deletion diagnostics.

•

The diagnostics are generalizations of Cook’s D, dfbeta, and leverage for 

general linear models and generalized linear models.

•

There are no published cutoffs for the GEE diagnostic statistics.

Leverage

•

Values range from 0 to 1.

•

Large leverage values might indicate that an observation or cluster is 

influential.

•

The sum of all leverage values is equal to p, the number of parameters in 

the model.

•

Values are produced for both observations and clusters.


